

PROGRAM

KAREL

IQI 151



řešené příklady

**PROGRAM KAREL**

**pro IQ 151**

**řešené příklady**

**Komenium, n.p., Praha**

**1986**

## Úvod

Text navazuje na návod "Program Karel pro IQ 151", vydaný n. p. Komenium v roce 1986, který popisoval činnost programu "Karel" a práci s programem dokumentoval na jednoduchých příkladech. Tento text doplňuje a rozšiřuje původní návod o některé další vesměs složitější řešené příklady. Z výše uvedeného důvodu se v příkladech budou vyskytovat některé jednoduché příkazy, které jsou uváděny v předchozím návodu. Jsou to:

a/ příkaz pro otočení robota čelem vzad bez akustických signálů /dále: pípání/, který má název CELEMVZAD a vypadá /viz úloha 1 předchozího návodu/:

CELEMVZAD  
ZNAMENA  
RYCHLE  
VLEVOVBOK  
VLEVOVBOK  
KONEC

b/ příkaz pro otočení robota vpravo vbok bez pípání, který má název VPRAVOVBOK a vypadá /viz úloha 4 předchozího návodu/:

VPRAVOVBOK  
ZNAMENA  
RYCHLE  
VLEVOVBOK  
VLEVOVBOK  
VLEVOVBOK  
KONEC

c/ příkaz, podle něhož dojde robot z libovolného místa k protilehlé zdi bez pípání. Příkaz má název KEZDI a vypadá:

KEZDI  
ZNAMENA  
DOKUD NEBUDE ZED  
RYCHLE  
KROK  
KONEC  
KONEC

Tento poslední příkaz je obdobný jako v úloze 3 předchozího návodu jen s tím rozdílem, že příkaz probíhá v režimu "RYCHLE", tedy bez signálů /pípání/. Pokud má tedy váš robot v paměti původní znění příkazu KEZDI, vymažte ho a nahraďte výše uvedeným.

Nyní následují jednotlivé řešené úlohy.



#### Úloha 1:

Vytvořte příkaz, podle něhož udělá robot krok v šikmém směru /tj. pod úhlem  $45^\circ$  od směru SEVER-JIH nebo VÝCHOD-ZÁPAD/ a přitom jednou pípne.

#### Řešení:

Robot udělá nejdříve pomalu krok /s pípnutím/ ve směru, do něhož je čelem otočen, pak udělá rychle /bez pípnutí/ vlevo v bok a krok v novém směru, nakonec se otočí bez pípnutí vpravo v bok. Celý příkaz, který pojmenujeme KROKS45 je následující:

```
KROKS45
ZNAMENA
POMALU
KROK
RYCHLE
VLEVOVBOK
KROK
VPRAVOVBOK
KONEC
```

Příkaz VPRAVOVBOK je popsán v úvodu. Nový příkaz vložíme v režimu PROGRAMOVÁNÍ a v režimu provádění pak ověříme, že robot provádí požadovanou činnost. Pozor pouze na vhodnou volbu výchozího postavení robota, aby při provádění příkazu KROKS45 nenarazil na zeď.



## Úloha 2:

Vytvořte příkaz, podle něhož robot přejde úhlopříčně prázdné město. Vyjde z libovolného rohu, přičemž má zeď za sebou a po pravici.

### Řešení:

Jedná se o několikanásobné opakování předchozího kroku v úhlopříčném směru, který popisuje předchozí úloha. Příkaz nazvaný UHLOPRICKA je následující:

```
UHLOPRICKA
ZNAMENA
DOKUD NEBJDE ZED
KROKS45
KONEC
KONEC
```

Příkaz vložíme v režimu PROGRAMOVÁNÍ, v režimu PROVÁDĚNÍ jeho činnost prověříme. Pro předchozí umístění robota do vhodné výchozí polohy použijeme v jednom z případů například vložení příkazu KAREL v režimu PROVÁDĚNÍ.



## Úloha 3:

Sestavte příkaz, pomocí něhož robot projde podél celé severní a jižní zdi města, po obou úhlopříčkách a na každém poli uvedených čtyř cest zanechá značku 1.

### Řešení:

Sestavovaný nový příkaz se bude jmenovat MOTYLEK, bude obsahovat dvakrát příkaz KEZDI+, který způsobí chod robota podél obou uvedených zdí města a dále dvakrát příkaz UHLOPRICKA+, který způsobí chod robota po úhlopříčkách. Oba dílčí příkazy končící znakem + jsou vlastně již dříve známé příkazy KEZDI a UHLOPRICKA /viz úvod a úloha 2/, doplněné pouze o příkaz k polo-

žení značky. V režimu PROGRAMOVÁNÍ vložíme:

KEZDI+	UHLOPRICKA+
ZNAMENA	ZNAMENA
DOKUD NEBUDE ZED	DOKUD NEBUDE ZED
RYCHLE	POLOZ
POLOZ	KROKS45
KROK	KONEC
KONEC	KONEC
KONEC	

MOTYLEK  
ZNAMENA  
KAREL  
KEZDI+  
VLEVOVBOK  
UHLOPRICKA+  
VPRAVOVBOK  
KEZDI+  
CELEMVZAD  
UHLOPRICKA+  
KONEC

Všimněte si, že výchozí postavení zajišťuje příkaz KAREL, dále se využívá i dříve zavedených jednoduchých příkazů VPRAVOVBOK a CELEMVZAD /viz úvod/. Zadaním příkazu MOTYLEK v režimu PROVÁDĚNÍ opět prověříme jeho správnost.

Poznámka:

Zdůvodněte sami, proč lze v příkazech KEZDI+ a UHLOPRICKA+ použít místo řádku

DOKUD NEBUDE ZED

řádek

OPAKUJ 11 KRAT

víme-li, že rozměry města jsou 12 x 12 kroků.



Úloha 4:

Vytvořte příkaz, pomocí něhož bude robot pípnutím ohlašovat každou asi desátou sekundu.

### Řešení:

Necháme robota otáčet na místě pomocí příkazů VLEVOVBOK, z nichž jsou tři provedeny rychle /bez pípnutí/ a čtvrtý je proveden pomalu /s pípnutím/. Do vytváření příkazu je nutno zařadit příkaz CEKEJLS, jehož prováděcí doba je asi 1 sekunda, přičemž jinak příkaz nemá žádný účinek. Toto sekundové zdržení je nutno desetkrát opakovat, aby robot pípnul vždy po deseti sekundách. Vytvářený příkaz se bude jmenovat PIP1ØPIP. V režimu PROGRAMOVÁNÍ vložíme následující příkazy:

PIP1ØPIP	CEKEJLS
ZNAMENA	ZNAMENA
RYCHLE	OPAKUJ 67 KRAT
VLEVOVBOK	KDYZ JE SEVER
VLEVOVBOK	KDYZ JE SEVER
VLEVOVBOK	KONEC JINAK
POMALU	KONEC
VLEVOVBOK	KONEC JINAK
OPAKUJ 1Ø KRAT	KONEC
CEKEJLS	KONEC
KONEC	KONEC
PIP1ØPIP	
KONEC	

Zadáme-li v režimu PROVÁDĚNÍ příkaz PIP1ØPIP, pak robot plní úlohou požadovanou funkci. Všimněte si, že tento program je ve tvaru nekonečného cyklu, který musí operátor zastavit pomocí tlačítka na klávesnici počítače.



### úloha 5:

Robot má udělat na místě piruetu /otočku o 360°/ doleva, čekat 5 s, udělat piruetu doprava, počkat 3 s a nakonec udělat piruetu doleva. Sestavte příkaz, pomocí něhož robot všechny uvedené činnosti vykoná.

### Řešení:

Příkaz pro provedení piruety vlevo se bude jmenovat

PIRJL, pro piruetu vpravo PIRJP. Pro pauzy použijeme několiknásobné opakování příkazu CEKEJLS z úlohy 4. Celkový požadovaný příkaz se bude jmenovat TOCENI. Využívá se rovněž příkazu VPRAVOVBOK z úvodu textu. V režimu PROGRAMOVÁNÍ vložíme následující příkazy:

TOCENI	PIRJL
ZNAMENA	ZNAMENA
PIRJL	OPAKUJ 4 KRAT
OPAKUJ 5 KRAT	VLEVOVBOK
CEKEJLS	KONEC
KONEC	KONEC
PIRJP	
OPAKUJ 3 KRAT	PIRUP
CEKEJLS	ZNAMENA
KONEC	OPAKUJ 4 KRAT
PIRJL	VPRAVOVBOK
KONEC	KONEC
	KONEC

Správnost funkce ověříme opět zadáním příkazu TOCENI v režimu PROVÁDĚNÍ.



#### Úloha 6:

Vytvořte příkaz, pomocí něhož robot zjistí, zda je nalevo od něj zeď. Pokud ano, má postoupit o krok dále podél zdi. Nemí-li nalevo od něj zeď, má pětkrát pípnout.

#### Řešení:

Příkaz pro pětinasobné pípnutí se bude nazývat 5XPIP a bude součástí celkového příkazu VLEVOZED. Využije se i příkazu VPRAVOVBOK z úvodu textu.

Při provádění příkazu VLEVOZED udělá robot nejdříve vlevo vbok a zjistí, zda je v této poloze před ním zeď. Pokud ano, otočí se zpět a udělá krok podél zdi. Pokud před ním zeď po prvním natočení nebude, provede příkaz 5XPIP. V režimu PROGRAMOVÁNÍ vložíme následující příkazy:

VLEVOZED  
ZNAMENA  
RYCHLE  
VLEVOVBOK  
KDYZ BUDE ZED  
VPRAVOVBOK  
KROK  
KONEC JINAK  
5XPIP  
KONEC  
KONEC

5XPIP  
ZNAMENA  
POMALU  
OPAKJJ 5 KRAT  
VLEVOVBOK  
KONEC  
RYCHLE  
VPRAVOVBOK  
KONEC

Zadáním příkazu VLEVOZED v režimu PROVÁDĚNÍ prověříme opět správnost jeho funkce.



### Úloha 7:

Robot je ze tří stran obklopen zdí. Vytvořte příkaz, pomocí něhož se otočí do směru, ve kterém není zeď a položením určeného počtu značek na své stanoviště označí světovou stranu, na níž je volná cesta. /SEVER - žádná značka, ZÁPAD - jedna, JIH - dvě, VÝCHOD - tři/.

### Řešení:

Úloha předpokládá předchozí vybudování vhodných stěn města a umístění robota tak, aby byl ze tří stran obklopen buď obvodovými nebo vnitřními zdmi a pouze v posledním čtvrtém směru aby zeď nebyla.

Výsledný příkaz se bude jmenovat PRUCHOD a bude v sobě zahrnovat dva jednodušší příkazy - NORD a HLEDAM.

Pomocí příkazu NORD se robot nejdříve otočí čelem na sever. Volnou cestu hledá pomocí příkazu HLEDAM tak, že pokud je čelem ke zdi, udělá vlevo vbok a současně položí značku. To opakuje tak dlouho, až je před ním volná cesta. Pak udělá tímto směrem jeden krok.

V režimu PROGRAMOVÁNÍ vložíme následující příkazy:

PRUCHOD  
ZNAMENA  
NORD  
HLEDAM  
KONEC

NORD  
ZNAMENA  
DOKUD NENI SEVER  
VLEVOVBOK  
KONEC  
KONEC

HLEDAM  
ZNAMENA  
KDYZ BUDE ZED  
VLEVOVBOK  
POLOZ  
HLEDAM  
KONEC JINAK  
KROK  
KONEC  
KONEC

Zadáním příkazu PRUCHOD v režimu PROVÁDĚNÍ při vhodně postavených zdech města a výchozí poloze robota ověříme jeho správnost.



#### Úloha 8:

Robot stojí v základní poloze. U západní zdi města je postavena zeď o velikosti jednoho tiskového místa a ve vzdálenosti lichého počtu tiskových míst od jižní zdi města. Vytvořte příkaz, pomocí něhož se robot přemístí do středu vzdálenosti mezi jižní zeď města a postavenou zeď.

#### Řešení:

Využijeme příkazů CELEMVZAD a KEZDI z úvodu textu. Robot udělá nejdříve ze základního postavení vlevo vbok, položí značku, dojde k postavené zdi a tam rovněž položí značku a po otočení čelem vzad udělá krok směrem k výchozímu místu. To nám /kromě prvního VLEVOVBOK provede příkaz POLKEZDI, který bude součástí hledaného výsledného příkazu.

Dále se robot podívá, zda těsně před ním není již položená značka. Pokud ano, zvedne značku před sebou i za sebou a vrátí se na místo, z něhož vyšel obě značky zvednout, protože je to tiskové pole uprostřed vzdále-

nosti obou zdí. Pokud nezjistí robot před sebou položenou značku, dojde až před položenou značku pomocí příkazu PREDZNACKU, na tomto místě položí další značku a původní zvedne. Totéž provede u druhé zdi. Tím vlastně zkrátí interval, v němž leží střed vzdálenosti obou zdí - proto se příkaz bude jmenovat ZKRACUJl. Automaticky v sobě tento příkaz zahrnuje příkaz PREDZNACKU. Výsledný příkaz nazveme STREDl, zahrnuje v sobě příkaz STOPSTREDl, který popsáním způsobem zkracuje vzdálenosti a zastavuje robota na středu vzdálenosti obou zdí. Před zahájením aktivity robota je nutno pomocí příkazu MESTO v režimu PROVÁDĚNÍ postavit vhodnou zeď, která se napojuje na západní zeď města a postavit robota do základní polohy příkazem KAREL. Pak v režimu PROGRAMOVÁNÍ vložíme následující příkazy:

POLPREDZDI  
 ZNAMENA  
 KDYZ BUDE ZED  
 KONEC JINAK  
 POLOZ  
 KEZDI  
 POLOZ  
 CELEMVZAD  
 KROK  
 KONEC  
 KONEC

PREDZNACKU  
 ZNAMENA  
 DOKUD NEB. ZNA.  
 KROK  
 KONEC  
 KONEC

STREDl  
 ZNAMENA  
 RYCHLE  
 VLEVOVBOK  
 POLPREDZDI  
 STOPSTREDl  
 KONEC

ZKRACUJ1  
ZNAMENA  
OPAKUJ 2 KRAT  
PREDZNACKU  
POLOZ  
KROK  
ZVEDNI  
CELEMVZAD  
KROK  
KONEC  
KONEC

STOPSTRED1  
ZNAMENA  
KDYZ BUDE ZNACKA  
KROK  
ZVEDNI  
CELEMVZAD  
KROK  
KROK  
ZVEDNI  
CELEMVZAD  
KROK  
KONEC JINAK  
ZKRACUJ1  
KROK  
STOPSTRED1  
KONEC  
KONEC

Zadáme-li v režimu PROVÁDĚNÍ příkaz STRED1, robot provede požadovanou činnost.



#### Úloha 9:

Robot stojí v základním postavení. Před ním je ve vzdálenosti 8 tiskových míst od západní zdi postavena zeď ve směru sever-jih a v ní v neznámém místě otvor o velikosti 1 tiskového pole. Sestavte příkaz, pomocí něhož robot tento otvor najde a zůstane v něm stát.

#### Řešení:

Využijeme příkazu VPRAVOVBOK z úvodu. Hledaný příkaz se bude jmenovat ZDEOTVOR, bude obsahovat dílčí příkazy 7KROKU a HLEDAME. Pomocí příkazu 7KROKU se robot přemístí ze základní polohy na osmé tiskové místo podél jižní zdi města. Nyní bude robot pokračovat v činnosti podle příkazu HLEDAME - podívá se, zda je před ním zeď. Pokud ne, udělá krok a zůstane stát, neboť je právě v průchodu. Pokud je před ním zeď, přesune se o jedno tiskové pole severněji a činnost v předchozím uvedenou opakuje, dokud otvor nenažde.

V režimu PROGRAMOVÁNÍ vložíme:

HLEDAME  
ZNAMENA  
KDYZ NEBUDE ZED  
KROK  
KONEC JINAK  
VLEVOVBOK  
KROK  
VPRAVOVBOK  
HLEDAME  
KONEC  
KONEC

7KROKU  
ZNAMENA  
OPAKUJ 7 KRAT  
KROK  
KONEC  
KONEC

ZDEOTVOR  
ZNAMENA  
KAREL  
7KROKU  
HLEDAME  
KONEC

V režimu PROVÁDĚNÍ nejdříve vytvoříme uvnitř města zeď odpovídající podmínkám a pak zadáme příkaz ZDEOTVOR. Robot provede požadovanou činnost.



Úloha 10:

/Varianta předchozí úlohy./

Zeď je postavena ve vzdálenosti ne delší než 8 polí.

Řešení:

Příkaz 7KROKU je nyní nepoužitelný. Robot musí při každém kroku k vnitřní zdi zjistit, zda již otvorem v této zdi právě neprochází. Zjistí to tak, že se natočí vlevo vbok a podívá se, zda na tiskovém poli před ním není konec postavené zdi. Pokud tam zeď je, robot se nachází právě v průchodu a svou aktivitu končí. Pokud tam zeď není, postoupí o jeden krok východním směrem. Dorazí-li k postavené zdi, analogickým způsobem jako v minulé úloze najde průchod a zůstane v něm stát. Výsledný příkaz se bude jmenovat OTVOR, bude obsahovat dílčí příkazy OTVORVEZDI, USZDISTOP a JVZDIOTOC. V režimu PROGRAMOVÁNÍ vložíme všechny jmenované příkazy v tomto tvaru:

OTVORVEZDI  
ZNAMENA  
KROK  
VPRAVOVBOK  
KDYZ NEBUDE ZED  
KROK  
KONEC JINAK  
VLEVOVBOK  
OTVORVEZDI  
KONEC  
KONEC

USZDISTOP  
ZNAMENA  
KDYZ BUDE ZED  
KONEC JINAK  
VPRAVOVBOK  
KROK  
UVZDIOTOC  
KONEC  
KONEC

UVZDIOTOC  
ZNAMENA  
KDYZ BUDE ZED  
VLEVOVBOK  
OTVORVEZDI  
KONEC JINAK  
VLEVOVBOK  
USZDISTOP  
KONEC  
KONEC

OTVOR  
ZNAMENA  
KAREL  
UVZDIOTOC  
KONEC

V režimu PROVÁDĚNÍ nejdříve pomocí příkazu MESTO postavíme zeď odpovídající podmínkám úlohy. Pak vložíme příkaz OTVOR a vidíme, že robot provede požadovanou činnost.



#### Úloha 11:

Sestavte příkaz, pomocí něhož robot položí značky 1 na trojúhelníkové ploše. Přepona trojúhelníka má být úhlopříčka města, vrchol s pravým úhlem je v levém spodním rohu města.

#### Řešení:

Nejdříve je nutno uvolnit plochu města od případných zdí nebo položených značek pomocí příkazu MESTO v režimu PROVÁDĚNÍ. Hledaný příkaz se bude jmenovat TROJUHELNIK a bude využívat jednodušších příkazů KEZDI<sup>^</sup>, SCHOD, STOPKROK, NAKONEC1 a TROJUHEL. V těchto jednodušších příkazech se vyskytují příkazy KEZDI, VPRAVOVBOK a CELEMVZAD z úvodu.

Robot položí nejdříve na výchozí postavení značku, pak pomocí KEZDI<sup>^</sup> označuje celou cestu podél jižní zdi. Dále podle příkazu SCHOD se přesune na severozápadní sousední pole a položí tam značku. /SCHOD je již součástí příkazu TROJUHEL !/ Pak se robot vrací k západní zdi a cestu značuje. Před západní zdí se otočí, přejde všechna označovaná pole této řady podle příkazu NAKONEC1 a již popsáním způsobem přejde na severozápadní sousední pole. Všimněte si, že příkaz STOPKROK jakožto součást příkazu SCHOD zastavuje činnost robota v okamžiku, kdy postoupil až k severní zdi města. V režimu PROGRAMOVÁNÍ vložíme následující příkazy:

```

KEZDI^
ZNAMENA
DOKUD NEBUDE ZED
RYCHLE
KROK
POLOZ
KONEC
KONEC

```

```

SCHOD
ZNAMENA
RYCHLE
VLEVOVBOK
STOPKROK
VLEVOVBOK
KROK
KONEC

```

```

TROJUHEL
ZNAMENA
SCHOD
POLOZ
KEZDI^
CELEMVZAD
NAKONEC1
VLEVOVBOK
KDYZ NEBUDE ZED
VPRAVOVBOK
TROJUHEL
KONEC JINAK
KONEC
KONEC

```

```

STOPKROK
ZNAMENA
KDYZ BUDE ZED
KONEC JINAK
KROK
KONEC
KONEC

```

```

NAKONEC1
ZNAMENA
DOKUD BUDE 1
RYCHLE
KROK
KONEC
KONEC

```

TROJUHLENIK  
ZNAMENA  
KAREL  
POLOZ  
KEZDI ^  
TROJUHLEL  
KONEC

Vložením příkazu TROJUHLENIK v režimu PROVÁDĚNÍ vytvoří robot požadovaný obrazec ze značek 1.



Úloha 12:

/Varianta předchozí úlohy./

Robot má vytvořený trojúhelník ze značek 1 po chvíli smazat.

Řešení:

Nejdříve je nutno uvolnit plochu města od značek a zdi z předchozích úloh. To provedeme pomocí příkazu MESTO v režimu PROVÁDĚNÍ. Hledaný příkaz se bude jmenovat TRCEKEJNIC, bude obsahovat příkaz pro vytvoření trojúhelníku TROJUHLENIK z předchozí úlohy, dále příkaz PAJZA způsobující čekání, nakonec příkaz SMAZANI, který vytvořený trojúhelník smaže. Příkaz SMAZANI obsahuje další příkazy SMAZ, KEZDI-, NAKONEC1-, OTACKAP. V těchto příkazech se opět využívají i příkazy VPRAVOVBOK, CELEMVZAD a KEZDI z úvodu, dále příkaz STOPKROK z předchozí úlohy. V režimu PROGRAMOVÁNÍ vložíme následující příkazy:

KEZDI-	NAKONEC1-
ZNAMENA	ZNAMENA
DOKUD NEBUDE ZED	DOKUD BUDE 1
RYCHLE	RYCHLE
KROK	KROK
ZVEDNI	ZVEDNI
KONEC	KONEC
KONEC	KONEC

OTACKAP  
ZNAMENA  
VPRAVOVBOK  
STOPKROK  
VPRAVOVBOK  
KONEC

PAUZA  
ZNAMENA  
OPAKUJ 99 KRAT  
OPAKUJ 2 KRAT  
KONEC  
KONEC

SMAZ  
ZNAMENA  
KEZDI  
OTACKAP  
ZVEDNI  
NAKONEC1 -  
CELEMVZAD  
VPRAVOVBOK  
KDYZ NEBUDE ZED  
VLEVOVBOK  
SMAZ  
KONEC JINAK  
KONEC  
KONEC

SMAZANI  
ZNAMENA  
KAREL  
ZVEDNI  
KEZDI -  
CELEMVZAD  
SMAZ  
KONEC

TRCEKEJNIC  
ZNAMENA  
TROJUHELNIK  
PAUZA  
SMAZANI  
KONEC

Pokuste se nyní již sami odvodit význam a působení jednotlivých zaváděných příkazů. O správnosti funkce příkazu TRCEKEJNIC se přesvědčíte jeho vložení v režimu PROVÁDĚNÍ.



### Úloha 13:

/Varianta předchozí úlohy./

Vytvořte příkaz, pomocí něhož robot postupně vytváří a maže 4 trojúhelníky, přičemž vrchol s pravým úhlem je nejdříve v dolním levém rohu, pak v dolním pravém rohu, pak v horním pravém rohu a nakonec v horním levém rohu.

### Řešení:

V režimu PROVÁDĚNÍ pomocí příkazu MESTO uvolníme plochu města od zdí nebo značek z předchozích úloh.

Pak v režimu PROGRAMOVÁNÍ vložíme následující příkazy:

START  
ZNAMENA  
CELEMVZAD  
KEZDI  
VLEVOVBOK  
KONEC

NASTART  
ZNAMENA  
CELEMVZAD  
KEZDI  
VLEVOVBOK  
KEZDI  
VLEVOVBOK  
KONEC

TROJL  
ZNAMENA  
POLOZ  
KEZDI^  
TROJHEL  
KONEC

SMAZANIL  
ZNAMENA  
ZVEDNI  
KEZDI-  
CELEMVZAD  
SMAZ  
KONEC

TRCEKEJNICI  
ZNAMENA  
TROJL  
PAUZA  
START  
SMAZANIL  
KONEC

TOCENITROJ  
ZNAMENA  
KAREL  
RYCHLE  
TRCEKEJNICI  
OPAKUJ 3 KRAT  
NASTART  
TRCEKEJNICI  
KONEC  
TOCENITROJ  
KONEC

Hledaný příkaz je TOCENITROJ, využívá dílčích příkazů TROJL, TRCEKEJNICI, SMAZANIL, START, NASTART. Z předchozích dvou úloh využívá příkazů KEZDI^, KEZDI-, SMAZ, PAUZA. Využívá se rovněž příkazů KEZDI a CELEMVZAD z úvodu.

Význam jednotlivých dílčích příkazů se již odvodíte snadno sami. O správnosti příkazu TOCENITROJ se přesvědčíte jeho vložení v režimu PROVÁDĚNÍ.

PROGRAM KAREL pro IQ 151 - řešené příklady  
Autor: Ing. Emil Kollert. Schválilo ministerstvo školství ČSR dne  
25.11.1985, č.j. 32 363/85-200 pro uživatele mikropočítače IQ 151.  
Vyrobitel Komenium, n.p., Praha pod ČKL 5025 v roce 1986. Odpověd-  
ná redaktorka: PhDr. Emilie Petráková. Obálka a technická úprava:  
Martina Mášová. Náklad: 1 500 ks. Vydání: 1.  
K-77/86

